



CarloSolo

SPOSTAMENTO CENTRO STELLA E IPOTESI SULLA CONDIZIONE DEL NEUTRO

19 January 2021

Il caso classico tratta lo spostamento del centro stella solo nel caso di interruzione del neutro, cioè neutro con impedenza infinita, in realtà passando da neutro con impedenza zero a neutro con impedenza infinita, che rappresenta l'interruzione dello stesso, la tensione V_0 tra centro astratto e centro concreto aumenta da zero fino al suo valore massimo se si verifica la condizione di carichi molto differenti sulle tre fasi. Se indichiamo con V_1, V_2, V_3 una terna di tensioni di fase simmetriche in partenza della linea, con I_1, I_2, I_3 le correnti sulle tre fasi assorbite dai carichi e con I_n la corrente sul neutro si ha:

$$I_1 + I_2 + I_3 = I_n$$

Si può quindi scrivere la seguente relazione:

$$Y_1 \cdot (V_1 - V_0) + Y_2 \cdot (V_2 - V_0) + Y_3 \cdot (V_3 - V_0) = Y_n \cdot V_0$$

per cui si ha :

-Teorema di Millman-

$$V_0 = (Y_1 \cdot V_1 + Y_2 \cdot V_2 + Y_3 \cdot V_3) / (Y_1 + Y_2 + Y_3 + Y_n)$$

A titolo di esempio ho ipotizzato un caso pratico che è il seguente:

Linea trifase BT in cavo aereo precordato di alluminio 3x70+1x54.6 di lunghezza pari a 0.6 Km che alimenta alla sua estremità tre carichi di valore differente distribuiti ognuno per fase. Le caratteristiche della linea sono riportate di seguito:

Tensione di fase in partenza $V_p = 230$ volt

Resistenza della fase $R_L = 0.443$ ohm/km

Reattanza della fase $X_L = 0.100$ ohm/km

Resistenza del neutro $R_n = 0.628$ ohm/km

Reattanza del neutro $X_n = 0.170$ ohm/km

Lunghezza linea $L = 0.6$ km

I carichi sono i seguenti:

Carico 1 kW 6.6 cos ϕ 0.90

Carico 2 kW 0.2 cos ϕ 0.90

Carico 3 kW 0.1 cos ϕ 0.95

Allo scopo di avere un riscontro pratico con valori numerici, si considerano tre condizioni di funzionamento della linea.

Ipotesi 1 - Neutro interrotto, impedenza infinita

Ipotesi 2 - Neutro con impedenza uguale a zero, (caso teorico)

Ipotesi 3 - Neutro con impedenza di valore finito, caso pratico sopra descritto con valore di impedenza definito.

Ho predisposto un programma in Fortran per poter calcolare agevolmente le tensioni e le correnti che circolano in linea nelle tre condizioni previste, il listato del programma è allegato a questo lavoro.

Nella ipotesi 1 sopra descritta sono impostati nel programma valori di resistenza e reattanza del neutro molto elevati che si avvicinano alla condizione di neutro interrotto, quindi si è posto una resistenza e reattanza del neutro di $1 \cdot 10^6$ ohm/km.

Nella ipotesi 2, neutro con impedenza uguale a zero, allo scopo di evitare l' overflow del calcolo si è impostato nel listato del programma un valore molto basso di resistenza e reattanza del neutro pari a $1 \cdot 10^{-6}$ ohm/Km.

Nella ipotesi 3 i valori di resistenza e reattanza del neutro sono quelli indicati nell'esempio.


Per ogni carico è stata calcolata la resistenza R_c e la reattanza X_c in base alla potenza e al cosfi dello stesso. Questi valori si suppongono costanti al variare della tensione. Si è quindi calcolata la resistenza R_t per ogni fase data dalla resistenza del carico R_c più la resistenza della fase di linea $R_L \cdot L$, quindi $R_t = R_c + R_L \cdot L$, dove L è la lunghezza di linea in km, stesso discorso vale per la reattanza quindi $X_t = X_c + X_L \cdot L$. Calcolate quindi le ammettenze delle fasi e del neutro si è calcolata la tensione $V_{0'0}$ tra il centro concreto ed il centro astratto e quindi anche le tensioni $V_{t10}, V_{t20}, V_{t30}$ come meglio indicato nello schema elettrico allegato.

Risulta di particolare interesse l'ipotesi 3 ove si nota che le tensioni V_{t20} e V_{t30} sono aumentate, in particolar modo la tensione sulla fase 3 è aumentata di circa 7 volt in conseguenza dello spostamento del centro stella dovuto al carico elevato sulla fase 1 rispetto ai carichi molto modesti sulle fasi 2 e 3.

Si sono infine calcolate le tensioni di fase sui carichi $V_{c1'0}, V_{c2'0}, V_{c3'0}$

Il programma salva nella stessa cartella ove è installato il file Risultati.txt che si riferisce alle 3 ipotesi considerate riguardo la condizione del neutro.

Allego i risultati dei calcoli nelle tre ipotesi considerate.



Schema elettrico.jpg

CALCOLO ESEGUITO CON NEUTRO INTERROTTO
IPOTESI 1

Resistenza di linea $R_L = 0.44$ ohm/Km
 Reattanza di linea $X_L = 0.10$ ohm/Km
 Lunghezza linea $L = 0.60$ Km
 Tensione di fase in partenza della linea
 $V_p = 230.00$ volt

Carico su fase 1 6.6 KW cosf 0.90
 Carico su fase 2 0.2 KW cosf 0.90
 Carico su fase 3 0.1 KW cosf 0.95

Tensione $V_{0'0}$ tra centro reale e centro astratto
 $V_{0'0} = 214.37 - 3.67$ volt
 Tensione $V_{0'0}$ in modulo
 $V_{0'0} = 214.41$ volt

Tensioni V_t in forma complessa
 $V_{t10} = 15.63 + 3.67$ volt
 $V_{t20} = -329.37 - 195.52$ volt
 $V_{t30} = -329.37 + 202.86$ volt

Tensioni V_t in modulo
 $V_{t10} = 16.05$ volt
 $V_{t20} = 383.03$ volt
 $V_{t30} = 386.83$ volt

Corrente sul neutro $I_0 = 0.00 - 0.00$ amp
 Modulo corrente sul neutro $I_0 = 0.00$ amp

Correnti I_c dei carichi in forma complessa
 $I_{c1} = 2.10 - 0.45$ amp
 $I_{c2} = -1.60 - 0.14$ amp
 $I_{c3} = -0.50 + 0.59$ amp

Correnti dei carichi I_c in modulo
 $I_{c1} = 2.15$ amp
 $I_{c2} = 1.61$ amp
 $I_{c3} = 0.77$ amp

Tensioni V_c sui carichi in forma complessa
 $V_{c1'0} = 15.04 + 3.66$
 $V_{c2'0} = -328.96 - 195.38$
 $V_{c3'0} = -329.21 + 202.73$

Tensioni V_c sui carichi in modulo
 $V_{c1'0} = 15.48$ volt
 $V_{c2'0} = 382.61$ volt
 $V_{c3'0} = 386.62$ volt

Ipotesi 1.PNG

CALCOLO ESEGUITO CON NEUTRO AD IMPEDENZA ZERO
IPOTESI 2

Resistenza di linea $R_L = 0.44$ ohm/Km
 Reattanza di linea $X_L = 0.10$ ohm/Km
 Lunghezza linea $L = 0.60$ Km
 Tensione di fase in partenza della linea
 $V_p = 230.00$ volt

Carico su fase 1 6.6 KW cosf 0.90
 Carico su fase 2 0.2 KW cosf 0.90
 Carico su fase 3 0.1 KW cosf 0.95

Tensione $V_{0'0}$ tra centro reale e centro astratto
 $V_{0'0} = 0.00$ 0.00 volt
 Tensione $V_{0'0}$ in modulo
 $V_{0'0} = 0.00$ volt

Tensioni V_t in forma complessa
 $V_{t10} = 230.00$ -0.00 volt
 $V_{t20} = -115.00$ -199.19 volt
 $V_{t30} = -115.00$ 199.19 volt

Tensioni V_t in modulo
 $V_{t10} = 230.00$ volt
 $V_{t20} = 230.00$ volt
 $V_{t30} = 230.00$ volt

Corrente sul neutro I_0 26.89 -13.27 amp
 Modulo corrente sul neutro I_0 29.99 amp

Correnti I_c dei carichi in forma complessa
 $I_{c1} = 27.79$ -13.17 amp
 $I_{c2} = -0.80$ -0.54 amp
 $I_{c3} = -0.09$ 0.45 amp

Correnti dei carichi I_c in modulo
 $I_{c1} = 30.75$ amp
 $I_{c2} = 0.97$ amp
 $I_{c3} = 0.46$ amp

Tensioni V_c sui carichi in forma complessa
 $V_{c1'0} = 221.82$ 1.83
 $V_{c2'0} = -114.82$ -198.99
 $V_{c3'0} = -114.95$ 199.07

Tensioni V_c sui carichi in modulo
 $V_{c1'0} = 221.83$ volt
 $V_{c2'0} = 229.74$ volt
 $V_{c3'0} = 229.88$ volt

Ipotesi 2.PNG

CALCOLO ESEGUITO CON NEUTRO LA CUI IMPEDENZA
E' DEFINITA DALL' UTENTE - IPOTESI 3

Resistenza di linea $R_L = 0.44$ ohm/Km
 Reattanza di linea $X_L = 0.10$ ohm/Km
 Resistenza del neutro $R_n = 0.63$ ohm/Km
 Reattanza del neutro $X_n = 0.17$ ohm/Km
 Lunghezza linea $L = 0.60$ Km
 Tensione di fase in partenza della linea
 $V_p = 230.00$ volt

Carico su fase 1 6.6 KW cosf 0.90
 Carico su fase 2 0.2 KW cosf 0.90
 Carico su fase 3 0.1 KW cosf 0.95

Tensione $V_{0'0}$ tra centro reale e centro astratto
 $V_{0'0} = 10.92 -2.04$ volt
 Tensione $V_{0'0}$ in modulo
 $V_{0'0} = 11.11$ volt

Tensioni V_t in forma complessa
 $V_{t10} = 219.08 \quad 2.04$ volt
 $V_{t20} = -125.92 -197.14$ volt
 $V_{t30} = -125.92 \quad 201.23$ volt

Tensioni V_t in modulo
 $V_{t10} = 219.09$ volt
 $V_{t20} = 233.93$ volt
 $V_{t30} = 237.38$ volt

Corrente sul neutro $I_0 \quad 25.64 \quad -12.36$ amp
 Modulo corrente sul neutro $I_0 \quad 28.46$ amp

Correnti I_c dei carichi in forma complessa
 $I_{c1} = 26.58 \quad -12.30$ amp
 $I_{c2} = -0.84 \quad -0.51$ amp
 $I_{c3} = -0.11 \quad 0.46$ amp

Correnti dei carichi I_c in modulo
 $I_{c1} = 29.29$ amp
 $I_{c2} = 0.98$ amp
 $I_{c3} = 0.47$ amp

Tensioni V_c sui carichi in forma complessa
 $V_{c1'0} = 211.28 \quad 3.72$
 $V_{c2'0} = -125.73 -196.96$
 $V_{c3'0} = -125.86 \quad 201.11$

Tensioni V_c sui carichi in modulo
 $V_{c1'0} = 211.31$ volt
 $V_{c2'0} = 233.67$ volt
 $V_{c3'0} = 237.25$ volt

Ipotesi 3.PNG

PROGRAM Sp_Centro_Stella
 ! Calcolo spostamento centro stella

```

IMPLICIT NONE
REAL:: Vp    ! tensione di fase in partenza in volt
REAL:: L     ! Lunghezza della linea in km
REAL:: Rn    ! resistenza del neutro in ohm/km
REAL:: Xn    ! reattanza del neutro in ohm/km
REAL:: RL   ! resistenza della fase in ohm/km
REAL:: XL   ! reattanza della fase in ohm/km
COMPLEX:: Yn ! ammettenza del neutro
COMPLEX:: V0 ! tensione stellata tra centro astratto e centro concreto
COMPLEX:: ZL ! impedenza della fase della linea in forma complessa
INTEGER:: i

REAL, DIMENSION(3):: P      !carico relativo alle fasi 1-2-3 in watt
REAL, DIMENSION(3):: cosf !fattore di potenza dei carichi fasi 1-2-3
REAL, DIMENSION(3):: arc  !arco in rad. corrisp.ai vari cosf
REAL, DIMENSION(3):: sinf !seno corrispondente ai cosf
REAL, DIMENSION(3):: Ic0  ! correnti nominali dei carichi
COMPLEX, DIMENSION(3):: Ic ! correnti dei carichi in forma complessa
COMPLEX:: I0              ! corrente sul neutro in forma complessa
REAL, DIMENSION(3):: Zc ! impedenza nominale dei carichi
REAL, DIMENSION(3):: Rc ! resistenza nominale dei carichi
REAL, DIMENSION(3):: Xc ! reattanza nominale dei carichi
REAL, DIMENSION(3):: Rt !somma tra la resistenza interna del carichi Rc
                        !e la resistenza della linea RL*L
REAL, DIMENSION(3):: Xt !somma tra la reattanza interna dei carichi Xc
                        !e la reattanza della linea XL*L
COMPLEX, DIMENSION(3):: Yt ! ammettenza totale carichi + linea
COMPLEX, DIMENSION(3):: V  ! tensioni di fase in partenza della linea
                        ! in forma complessa
COMPLEX, DIMENSION(3):: Vt ! tensioni totali su impedenza carico + linea
COMPLEX, DIMENSION(3):: Vc ! tensioni sui carichi
INTEGER:: sel

OPEN (UNIT=3, FILE='Risultati.txt', STATUS='REPLACE', ACTION='WRITE')

WRITE(*, "(' Inserire la tensione di fase in partenza della linea' &
        & ' in volt')")
READ(*, *) Vp
WRITE(*, "(' Inserire la resistenza di fase della linea in ohm/km')")
READ(*, *) RL
WRITE(*, "(' Inserire la reattanza di fase della linea in ohm/km')")
READ(*, *) XL

sel= -1

```

```

DO WHILE(sel<1 .OR. sel>3)
WRITE(*,"(' Selezionare lo stato di funzionamento del neutro',/ &
& ' 1) Neutro interrotto',/ &
& ' 2) Neutro con impedenza uguale a zero',/ &
& ' 3) Neutro con valori di impedenza definiti dall''utente'')")
READ(*,*) sel
SELECT CASE(sel)
CASE(1)
Rn=1E6
Xn=1E6
CASE(2)
Rn=1E-6
Xn=1E-6
CASE(3)
WRITE(*,"(' Inserire la resistenza del neutro della linea in ohm/km'")")
READ(*,*) Rn
WRITE(*,"(' Inserire la reattanza del neutro della linea in ohm/km'")")
READ(*,*) Xn
END SELECT
END DO

L= -1
DO WHILE(L<=0)
WRITE(*,"(' Inserire la lunghezza della linea in km'")")
READ(*,*) L
IF (L<=0) THEN
WRITE(*,"(' La lunghezza della linea deve essere maggiore di',/ &
& ' zero',/ )")
END IF
END DO

DO i= 1,3
WRITE(*,"(' Inserire il carico relativo alla fase ',I1,' in KW'")")i
READ(*,*) P(i)
WRITE(*,"(' Inserire il fattore di potenza relativo alla fase ',I1'")")i
READ(*,*) cosf(i)
END DO

DO i= 1,3
arc(i)= ACOS (cosf(i))
sinf(i)= SIN(arc(i))
END DO

```

```

DO i= 1,3
  Ic0(i) = (P(i)*1000)/(Vp*cosf(i))
END DO

DO i=1,3
  Zc(i)=Vp/Ic0(i)
  Rc(i)=Zc(i)*cosf(i)
  Xc(i)=Zc(i)*sinf(i)
END DO

DO i=1,3
  Rt(i)= Rc(i)+RL*L
  Xt(i)= Xc(i)+XL*L
  Yt(i)= 1/CMPLX(Rt(i),Xt(i))
END DO

IF ((Rn == 0.).AND.(Xn == 0.)) THEN
  WRITE(*, "(' Se Rn=0 e Xn=0 sono uguali a zero',/ &
    & ' viene impostato Rn=1E-6 e Xn=1E-6',/ &
    & ' per non mandare il calcolo in overflow',/)'")
  READ(*,*)
  Rn=1E-6
  Xn=1E-6
END IF

Yn= 1/CMPLX(Rn*L,Xn*L)

WRITE(*, "(//////////)")

IF (sel==1) THEN
  WRITE(3, "(' CALCOLO ESEGUITO CON NEUTRO INTERROTTO',/ , &
    & ' IPOTESI 1',/)'")
  WRITE(3, "(' Resistenza di linea RL= ',F4.2,A)")RL, ' ohm/Km'
  WRITE(3, "(' Reattanza di linea XL= ',F4.2,A)")XL, ' ohm/Km'
  WRITE(3, "(' Lunghezza linea L= ',F4.2,A)")L, ' Km'
ELSE IF (sel==2) THEN
  WRITE(3, "(' CALCOLO ESEGUITO CON NEUTRO AD IMPEDENZA ZERO',/ , &
    & ' IPOTESI 2',/)'")
  WRITE(3, "(' Resistenza di linea RL= ',F4.2,A)")RL, ' ohm/Km'
  WRITE(3, "(' Reattanza di linea XL= ',F4.2,A)")XL, ' ohm/Km'
  WRITE(3, "(' Lunghezza linea L= ',F4.2,A)")L, ' Km'
ELSE IF (sel==3) THEN
  WRITE(3, "(' CALCOLO ESEGUITO CON NEUTRO LA CUI IMPEDENZA ',/ , &

```



```

        & ' E'' DEFINITA DALL'' UTENTE - IPOTESI 3',/)"
WRITE(3,"(' Resistenza di linea   RL= ',F4.2,A)")RL,' ohm/Km'
WRITE(3,"(' Reattanza di linea   XL= ',F4.2,A)")XL,' ohm/Km'
WRITE(3,"(' Resistenza del neutro Rn= ',F4.2,A)")Rn,' ohm/Km'
WRITE(3,"(' Reattanza del neutro Xn= ',F4.2,A)")Xn,' ohm/Km'
WRITE(3,"(' Lunghezza linea      L= ',F4.2,A)")L,' Km'
END IF

WRITE(3,"(' Tensione di fase in partenza della linea',/, &
        & ' Vp =',F8.2,A,/)")Vp,'volt'

DO i= 1,3
    WRITE(3,"(' Carico su fase '&
            & ',I1,F8.1,' KW',' cosf ',F4.2 )")i,P(i),cosf(i)
END DO

! Tensioni di fase in partenza della linea in forma complessa
V(1)= CMPLX (Vp,0.)
V(2)= CMPLX (-Vp * 0.5 , -Vp * SQRT(3.)/2)
V(3)= CMPLX (-Vp * 0.5 , Vp * SQRT(3.)/2)

V0= (V(1)*Yt(1)+V(2)*Yt(2)+V(3)*Yt(3))/(Yt(1)+Yt(2)+Yt(3)+Yn)

WRITE(3,"(/' Tensione V0''0 tra centro reale e centro astratto ',/,&
        & ' V0''0 =',2F8.2,A)")V0,' volt'
WRITE(3,"(' Tensione V0''0 in modulo',/,&
        & ' V0''0 =',F8.2,A,/)") SQRT(REAL(V0)**2+AIMAG(V0)**2),' volt'

WRITE(3,"(' Tensioni Vt in forma complessa'")
DO i= 1,3
    Vt(i)= V(i)-V0
    WRITE(3,"(' Vt',I1,'0 =',2F8.2,A)")&
        & i,Vt(i),' volt'
END DO

WRITE(3,"(/, ' Tensioni Vt in modulo'")
DO i= 1,3
    WRITE(3,"(' Vt',I1,'0 =',F8.2,A)") &
        & i, SQRT(REAL(Vt(i))**2+AIMAG(Vt(i))**2),' volt'
END DO

I0 = Yn*V0
WRITE(3,"(/, ' Corrente sul neutro I0',2F8.2,A)")I0,' amp'

```

```

WRITE(3, "(' Modulo corrente sul neutro I0 ',F8.2,A,/)") &
      &   SQRT (REAL(I0)**2+AIMAG(I0)**2), ' amp'

ZL=CMPLX(RL*L,XL*L)

WRITE(3, "(' Correnti Ic dei carichi in forma complessa' )")
DO i = 1,3
  Ic(i) = Yt(i)*Vt(i)
  WRITE(3, "(' Ic',I1, ' =',2F8.2,A) ")i,Ic(i), ' amp'
END DO

WRITE(3, ("(/, ' Correnti dei carichi Ic in modulo' )")
DO i = 1,3
  WRITE(3, ("(' Ic',I1, ' =',F8.2,A) ") &
      & i, SQRT (REAL (Ic(i))**2+AIMAG(Ic(i))**2), ' amp'
END DO

WRITE(3, ("(/, ' Tensioni Vc sui carichi in forma complessa' )")
DO i= 1,3
  Vc(i)= Vt(i)-(ZL*Ic(i))
  WRITE(3, ("(' Vc',I1, ' =',2F8.2) ")i,Vc(i)
END DO

WRITE(3, ("(/, ' Tensioni Vc sui carichi in modulo' )")
DO i= 1,3
  WRITE(3, ("(' Vc',I1, ' =',F8.2,A) ") &
      & i, SQRT (REAL (Vc(i))**2+AIMAG (Vc(i))**2), ' volt'
END DO
CLOSE(3)

WRITE(*, ("(' Per visualizzare i risultati del calcolo apri il file',/ &
      & ' Risultati.txt, premi invio per continuare' )")
READ(*,*)
STOP
END PROGRAM Sp_Centro_Stella

```

Estratto da ["https://www.electroyou.it/mediawiki/index.php?title=UsersPages:CarloSolo:spostamento-centro-stella"](https://www.electroyou.it/mediawiki/index.php?title=UsersPages:CarloSolo:spostamento-centro-stella)